

NAG Fortran Library Routine Document

G13BGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G13BGF accepts a series of new observations of an output time series and any associated input time series, for which a multi-input model is already fully specified, and updates the 'state set' information for use in constructing further forecasts.

The previous specification of the multi-input model will normally have been obtained by using G13BEF to estimate the relevant transfer function and ARIMA parameters. The supplied state set will originally have been produced by G13BEF (or possibly G13BJF), but may since have been updated by G13BGF.

2 Specification

```

SUBROUTINE G13BGF(STTF, NSTTF, MR, NSER, MT, PARA, NPARA, NNV, XXYN,
1                IXXYN, KZEF, RES, WA, IWA, IFAIL)
      INTEGER      NSTTF, MR(7), NSER, MT(4,NSER), NPARA, NNV, IXXYN,
1                KZEF, IWA, IFAIL
      real        STTF(NSTTF), PARA(NPARA), XXYN(IXXYN,NSER), RES(NNV),
1                WA(IWA)

```

3 Description

The multi-input model is specified in Section 3 of the document for G13BEF. The form of these equations required to update the state set is as follows:

$$z_t = \delta_1 z_{t-1} + \delta_2 z_{t-2} + \cdots + \delta_p z_{t-p} + \omega_0 x_{t-b} - \omega_1 x_{t-b-1} - \cdots - \omega_q x_{t-b-q}$$

the transfer models which generate input component values $z_{i,t}$ from one or more inputs $x_{i,t}$,

$$n_t = y_t - z_{1,t} - z_{2,t} - \cdots - z_{m,t}$$

which generates the output noise component from the output y_t and the input components, and

$$\begin{aligned}
 w_t &= \nabla^d \nabla_s^D n_t - c \\
 e_t &= w_t - \Phi_1 w_{t-s} - \Phi_2 w_{t-2 \times s} - \cdots - \Phi_P w_{t-P \times s} + \Theta_1 e_{t-s} + \Theta_2 e_{t-2 \times s} + \cdots + \Theta_Q e_{t-Q \times s} \\
 a_t &= e_t - \phi_1 e_{t-1} - \phi_2 e_{t-2} - \cdots - \phi_p e_{t-p} + \theta_1 a_{t-1} + \theta_2 a_{t-2} + \cdots + \theta_q a_{t-q}
 \end{aligned}$$

the ARIMA model for the output noise which generates the residuals a_t .

The state set (as also given in Section 3 of the document for G13BEF) is the collection of terms

$$z_{n+1-k}, x_{n+1-k}, n_{n+1-k}, w_{n+1-k}, e_{n+1-k} \quad \text{and} \quad a_{n+1-k}$$

for $k = 1$ up to the maximum lag associated with each of these series respectively, in the above model equations. n is the latest time point of the series from which the state set has been generated.

The routine accepts further values of the series $y_t, x_{1,t}, x_{2,t}, \dots, x_{m,t}$ for $t = n + 1, n + 2, \dots, n + l$, and applies the above model equations over this time range, to generate new values of the various model components, noise series and residuals. The state set is reconstructed, corresponding to the latest time point $n + l$, the earlier values being discarded.

The set of residuals corresponding to the new observations may be of use in checking that the new observations conform to the previously fitted model. The components of the new observations of the output series which are due to the various inputs, and the noise component, are also optionally returned.

The parameters of the model are not changed in this routine.

4 References

Box G E P and Jenkins G M (1976) *Time Series Analysis: Forecasting and Control* (Revised Edition) Holden-Day

5 Parameters

- 1: STTF(NSTTF) – *real* array *Input/Output*
On entry: the NSTTF values in the state set before updating as returned by G13BEF, G13BJF or a previous call to G13BGF.
On exit: the state set values after updating.

- 2: NSTTF – INTEGER *Input*
On entry: the exact number of values in the state set array STTF as returned by G13BEF or G13BJF.

- 3: MR(7) – INTEGER array *Input*
On entry: the orders vector (p, d, q, P, D, Q, s) of the ARIMA model for the output noise component.
 p, q, P and Q refer respectively to the number of autoregressive (ϕ), moving average (θ), seasonal autoregressive (Φ) and seasonal moving average (Θ) parameters.
 d, D and s refer respectively to the order of non-seasonal differencing, the order of seasonal differencing, and the seasonal period.

- 4: NSER – INTEGER *Input*
On entry: the total number of input and output series. There may be any number of input series (including none), but only one output series.

- 5: MT(4,NSER) – INTEGER array *Input*
On entry: the transfer function model orders b, p and q of each of the input series. The data for input series i are held in column i . Row 1 holds the value b_i , row 2 holds the value q_i and row 3 holds the value p_i . For a simple input, $b_i = q_i = p_i = 0$.
Row 4 holds the value r_i , where $r_i = 1$ for a simple input and $r_i = 2$ or 3 for a transfer function input. When $r_i = 1$ any non-zero contents of rows 1, 2 and 3 of column i are ignored. The choice of $r_i = 2$ or $r_i = 3$ is an option for use in model estimation and does not affect the operation of this routine.
Constraint: $MT(4, i) = 1, 2$ or 3 for $i = 1, 2, \dots, NSER - 1$.

- 6: PARA(NPARA) – *real* array *Input*
On entry: estimates of the multi-input model parameters as returned by G13BEF. These are in order, firstly the ARIMA model parameters: p values of ϕ parameters, q values of θ parameters, P values of Φ parameters and Q values of Θ parameters. These are followed by the transfer function model parameter values $\omega_0, \omega_1, \dots, \omega_{q_1}, \delta_1, \delta_2, \dots, \delta_{p_1}$ for the first of any input series and similarly for each subsequent input series. The final component of PARA is the value of the constant c .

- 7: NPARA – INTEGER *Input*
On entry: the exact number of $\phi, \theta, \Phi, \Theta, \omega, \delta$ and c parameters. (c must be included whether its value was previously estimated or was set fixed.)

- 8: NNV – INTEGER *Input*
On entry: the number of new observation sets being used to update the state set, each observation set consisting of a value of the output series and the associated values of each of the input series at a particular time point.
- 9: XXYN(IXXYN,NSER) – *real* array *Input/Output*
On entry: the NNV new observation sets being used to update the state set. Column i contains the values of input series i . Column NSER contains the values of the output series. Consecutive rows correspond to increasing time sequence.
On exit: if KZEF = 0, XXYN remains unchanged.
 If KZEF \neq 0, the columns of XXYN hold the corresponding values of the input component series z_t and the output noise component n_t , in that order.
- 10: IXXYN – INTEGER *Input*
On entry: the first dimension of the array XXYN as declared in the (sub)program from which G13BGF is called.
Constraint: IXXYN \geq NNV.
- 11: KZEF – INTEGER *Input*
On entry: KZEF must not be set to 0, if the values of the input component series z_t and the values of the output noise component n_t are to overwrite the contents of XXYN on exit, and must be set to 0 if XXYN is to remain unchanged on exit.
- 12: RES(NNV) – *real* array *Output*
On exit: the values of the residual series a_t corresponding to the new observations of the output series.
- 13: WA(IWA) – *real* array *Workspace*
 14: IWA – INTEGER *Input*
On entry: the dimension of the array WA as declared in the (sub)program from which G13BGF is called.
Constraint: IWA \geq NNV + 2 \times NSTTF + max(NNV, NSTTF) + max(NNV, ncc), where $ncc = 4 \times (p + q + P + Q)$.
- 15: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, NSTTF is not consistent with the orders in arrays MR and MT.

IFAIL = 2

On entry, NPARA is not consistent with the orders in arrays MR and MT.

IFAIL = 3

On entry, IXXYN is too small.

IFAIL = 4

On entry, IWA is too small.

IFAIL = 5

On entry, one of the r_i , stored in $MT(4, i)$, for $i = 1, 2, \dots, NSER - 1$ does not equal 1, 2 or 3.

7 Accuracy

The computations are believed to be stable.

8 Further Comments

The time taken by the routine is approximately proportional to $NNV \times NPARA$.

9 Example

The example uses the data described in G13BEF in which 40 observations of an output time series and a single input series were processed. In this example a model which included seasonal differencing of order 1 was used. The 10 values of the state set and the 5 final values of PARA then obtained are used as input to this program, together with the values of 4 new observations and the transfer function orders of the input series. The model used is $\phi_1 = 0.5158$, $\Theta_1 = 0.9994$, $\omega_0 = 8.6343$, $\delta_1 = 0.6726$, $c = -0.3172$.

The following are computed and printed out: the updated state set, the residuals a_t and the values of the components z_t and the output noise component n_t corresponding to the new observations.

9.1 Program Text

```
*      G13BGF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER      NSERMX, NPMAX, NSTTFM, NNVMAX, IXXYN, IWA
      PARAMETER    (NSERMX=2, NPMAX=10, NSTTFM=20, NNVMAX=40,
+                IXXYN=NNVMAX, IWA=160)
      INTEGER      NIN, NOUT
      PARAMETER    (NIN=5, NOUT=6)
*      .. Local Scalars ..
      INTEGER      I, IFAIL, J, KZEF, NNV, NPARA, NSER, NSTTF
*      .. Local Arrays ..
      real         PARA(NPMAX), RES(NNVMAX), STTF(NSTTFM), WA(IWA),
+                XXYN(IXXYN, NSERMX)
      INTEGER      MR(7), MT(4, NSERMX)
*      .. External Subroutines ..
      EXTERNAL     G13BGF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G13BGF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) NSTTF, NSER, NNV, KZEF
      IF (NSTTF.GT.0 .AND. NSTTF.LE.NSTTFM .AND. NSER.GT.0 .AND.
+      NSER.LE.NSERMX .AND. NNV.GT.0 .AND. NNV.LE.NNVMAX) THEN
      READ (NIN,*) (MR(I), I=1, 7)
      DO 20 I = 1, 4
          READ (NIN,*) (MT(I, J), J=1, NSER)
20      CONTINUE
```

```

      NPARA = 0
      DO 40 I = 1, NSER
        NPARA = NPARA + MT(2,I) + MT(3,I)
40    CONTINUE
      NPARA = NPARA + MR(1) + MR(3) + MR(4) + MR(6) + NSER
      READ (NIN,*) (STTF(I),I=1,NSTTF)
      IF (NPARA.LE.NPMA) THEN
        READ (NIN,*) (PARA(I),I=1,NPARA)
        DO 60 I = 1, NNV
          READ (NIN,*) (XXYN(I,J),J=1,NSER)
60    CONTINUE
          IFAIL = 0
*
          CALL G13BGF(STTF,NSTTF,MR,NSER,MT,PARA,NPARA,NNV,XXYN,IXXYN,
+             KZEF,RES,WA,IWA,IFAIL)
*
          WRITE (NOUT,*)
          WRITE (NOUT,*) 'The updated state set'
          WRITE (NOUT,99999) (STTF(I),I=1,NSTTF)
          WRITE (NOUT,*)
          WRITE (NOUT,*) 'The residuals (after differencing)'
          DO 80 I = 1, NNV
            WRITE (NOUT,99998) I, RES(I)
80    CONTINUE
          WRITE (NOUT,*)
          WRITE (NOUT,*) 'The values of z(t) and n(t)'
          DO 100 I = 1, NNV
            WRITE (NOUT,99998) I, (XXYN(I,J),J=1,NSER)
100   CONTINUE
          END IF
        END IF
      STOP
*
99999 FORMAT (1X,6F10.4)
99998 FORMAT (1X,I4,2F10.4)
      END

```

9.2 Program Data

G13BGF Example Program Data

```

10    2    4    1
1    0    0    0    1    1    4
1    0
0    0
1    0
3    0
6.0530 184.4749 -80.0885 -75.1704 -76.9481 -81.4749 0.7776 -2.6190
-2.3054 -1.1963
0.5158 0.9994 8.6343 0.6726 -0.3172
5.9410 96.0000
5.3860 95.0000
5.8110 80.0000
6.7160 88.0000

```

9.3 Program Results

G13BGF Example Program Results

The updated state set

6.7160	158.3155	-80.3412	-74.9035	-80.7814	-70.3155
0.8416	-2.0333	-5.8201	10.2810		

The residuals (after differencing)

1	1.4586
2	-2.4674
3	-4.7714
4	13.2830

The values of $z(t)$ and $n(t)$

1	176.3412	-80.3412
2	169.9035	-74.9035
3	160.7814	-80.7814
4	158.3155	-70.3155
